Álgebra Lineal con wxMaxima

1 Sistemas de ecuaciones lineales

Usar la pestaña Resolver

Ejemplo 1: x + 2*y = 5 , -x + y =1

```
(%i1) solve([x + 2*y = 5  , -x + y =1], [x,y]);
```

(%o1) $[[x=1,y=2]]$

Ejemplo 2: x + y + z = 5, x + 2*y + 3*z = 8

```
(%i2) solve([x + y + z = 5,   x + 2*y + 3*z = 8], [x, y, z]);
```

(%o2) $[[x=\%r1+2,y=3-2\,\%r1,z=\%r1]]$

observar que al ser

nº incógnitas > nº ecuaciones

la solución depende de un parámetro...

2 Operar con matrices

Introducir las matrices con la pestaña

Álgebra --> Introducir matriz

NOTAS:

- la multiplicación de matrices se escribe A.B (NO A*B)
- las potencias se escriben A^^2 (NO A^2)

Ejemplo 1: operar las matrices A y B del ejercicio 5

```
(%i3) A: matrix([1,-1], [1,1]);
```

(%o3) $\begin{bmatrix} 1 & -1 \\ 1 & 1 \end{bmatrix}$

```
(%i4) A^^2;A^^4;A^^(-2);
```

(%o4) $\begin{bmatrix} 0 & -2 \\ 2 & 0 \end{bmatrix}$

(%o5) $\begin{bmatrix} -4 & 0 \\ 0 & -4 \end{bmatrix}$

(%o6) $\begin{bmatrix} 0 & \frac{1}{2} \\ -\frac{1}{2} & 0 \end{bmatrix}$

(%i7) B: matrix([2,1],[4,2],[0,-1]);

(%o7) $\begin{bmatrix} 2 & 1 \\ 4 & 2 \\ 0 & -1 \end{bmatrix}$

(%i8) B.A;

(%o8) $\begin{bmatrix} 3 & -1 \\ 6 & -2 \\ -1 & -1 \end{bmatrix}$

(%i9) B.transpose(B);transpose(B).B;

(%o9) $\begin{bmatrix} 5 & 10 & -1 \\ 10 & 20 & -2 \\ -1 & -2 & 1 \end{bmatrix}$

(%o10) $\begin{bmatrix} 20 & 10 \\ 10 & 6 \end{bmatrix}$

NOTA: Observar qué ocurre si escribo A^2 (en vez A^^2)

(%i11) A^2;A^^2;

(%o11) $\begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}$

(%o12) $\begin{bmatrix} 0 & -2 \\ 2 & 0 \end{bmatrix}$

lo mismo con A*A (en vez de A.A)

(%i13) A.A;A*A;

(%o13) $\begin{bmatrix} 0 & -2 \\ 2 & 0 \end{bmatrix}$

(%o14) $\begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}$

EJEMPLO 2: Cálculo de determinantes, inversas,...

--> usar la pestaña Álgebra

(%i15) determinant(A);

(%o15) 2

(%i16) invert(A);

(%o16) $\begin{bmatrix} \frac{1}{2} & \frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{bmatrix}$

3 Autovalores y diagonalización

EJEMPLO 3: Cálculo de autovalores y autovectores

--> usar pestaña Álgebra

Ejercicio 7

```
(%i17) C: matrix([1,0,0], [-1,2,0], [2,0,-1]);
```

(%o17) $\begin{bmatrix} 1 & 0 & 0 \\ -1 & 2 & 0 \\ 2 & 0 & -1 \end{bmatrix}$

a) pestaña valores propios

```
(%i18) eigenvalues(C);
(%o18) [[-1,1,2],[1,1,1]]
```

los autovalores son {-1, 1, 2}, todos ellos simples
(esto es lo que indica la lista [1,1,1])

b) pestaña vectores propios

```
(%i19) eigenvectors(C);
(%o19) [[[-1,1,2],[1,1,1]],[[[0,0,1]],[[1,1,1]],[[0,1,0]]]]
```

NOTA: - el primer bloque son los autovalores [-1,1,2] (todos simples)
- autovector de x = -1 --> [0,0,1]
- autovector de x = 1 --> [1,1,1]
- autovector de x = 2 --> [0,1,0]

c) diagonalizar la matriz: no hay pestaña para esto, hay que hacerlo a mano.
- definir la matriz P (columnas de autovectores)
- definir la matriz diagonal D (se puede usar la pestaña tipo -> diagonal)
- comprobar que A = P.D.P^^(-1)

```
(%i20) P: matrix([0,1,0], [0,1,1], [1,1,0]);
```

(%o20) $\begin{bmatrix} 0 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{bmatrix}$

```
(%i21) D: matrix([-1,0,0], [0,1,0], [0,0,2]);
```

(%o21) $\begin{bmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 2 \end{bmatrix}$

```
(%i22) P.D.P^^(-1);
```

(%o22) $\begin{bmatrix} 1 & 0 & 0 \\ -1 & 2 & 0 \\ 2 & 0 & -1 \end{bmatrix}$

c) calcular A^n = P. D^n.P^^(-1)

(%i23) D^n;

(%o23) $\begin{bmatrix} (-1)^n & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 2^n \end{bmatrix}$

(%i24) An:P. D^n.P^^(-1);

(%o24) $\begin{bmatrix} 1 & 0 & 0 \\ 1-2^n & 2^n & 0 \\ 1-(-1)^n & 0 & (-1)^n \end{bmatrix}$

4 Sistemas Dinámicos Discretos

Aunque Excel es un poco más sencillo,
las recurrencias también pueden definirse y representarse en Maxima.

NOTA: Las sucesiones se definen con

x[n]:= ...

EJEMPLO 1: gatos y ratones

x[n+1] = 3*x[n] - y[n]
y[n+1] = x[n] + y[n]

con valores iniciales x[0]=10, y[0]=2

Paso 1: definir los valores iniciales (como ctes)

(%i25) x[0]:10;y[0]:2;
(%o25) 10
(%o26) 2

Paso 2: definir la recurrencia (como sucesiones)

OJO: no puedo definir x[n+1], debe ser x[n]

(%i27) x[n]:=3*x[n-1]-y[n-1];y[n]:=x[n-1]+y[n-1];
(%o27) $x_n := 3\, x_{n-1} - y_{n-1}$
(%o28) $y_n := x_{n-1} + y_{n-1}$

puedo ver valores particulares sustituyendo n

(%i29) x[4]; y[4];
(%o29) 416
(%o30) 288

Paso 3: para ver una lista de resultados hay que crearla con "makelist"
(conviene ponerle un nombre)

```
(%i31) raton:makelist([n,x[n]],n,0,8);gato:makelist([n,y[n]],n,0,8);
(%o31) [[0,10],[1,28],[2,72],[3,176],[4,416],[5,960],[6,2176],[7,
4864],[8,10752]]
(%o32) [[0,2],[1,12],[2,40],[3,112],[4,288],[5,704],[6,1664],[7,
3840],[8,8704]]
```

Paso 4: para dibujarla usar dentro de plot2d el comando "discrete"

```
(%i33) wxplot2d([[discrete,raton],[discrete,gato]], [legend, ratones, gatos])
```

(%t33)



(%o33)

Paso 5: crear listas similares para visualizar la tasa de crecimiento, etc..

```
(%i34) fpprintprec:3;
(%o34) 3
```

```
(%i35) tasa:makelist([x[n+1]/x[n],y[n+1]/y[n]], n,0,20),numer;
(%o35) [[2.8,6],[2.57,3.33],[2.44,2.8],[2.36,2.57],[2.31,2.44],[
2.27,2.36],[2.24,2.31],[2.21,2.27],[2.19,2.24],[2.17,2.21],[2.16,
2.19],[2.15,2.17],[2.14,2.16],[2.13,2.15],[2.12,2.14],[2.11,2.13],[
2.11,2.12],[2.1,2.11],[2.1,2.11],[2.09,2.1],[2.09,2.1]]
```

Se observa que a largo plazo las poblaciones se multiplican aprox por 2.
También puede sustituirse en un valor alto

```
(%i36) x[101]/x[100],numer;
(%o36) 2.02
```

```
(%i37) proporc:makelist([x[n]/(x[n]+y[n]),y[n]/(x[n]+y[n])], n,0,20),numer;
(%o37) [[0.8,0.2],[0.7,0.3],[0.6,0.4],[0.6,0.4],[0.6,0.4],[0.6,0.4
],[0.6,0.4],[0.6,0.4],[0.6,0.4],[0.5,0.5],[0.5,0.5],[0.5,0.5],[0.5
,0.5],[0.5,0.5],[0.5,0.5],[0.5,0.5],[0.5,0.5],[0.5,0.5],[0.5,0.5],
[0.5,0.5],[0.5,0.5]]
```

Se observa que las poblaciones tienden a distribuirse 50% ratones y 50% gato
(aunque en la gráfica se ve que SIEMPRE hay más ratones que gatos).

Paso 6: La matriz del sistema debería confirmar este hecho,
con autovalor dominante 2 y autovector (0.5,0.5)

```
(%i38) A: matrix([3,-1], [1,1]);
```

$$(\%o38) \quad \begin{bmatrix} 3 & -1 \\ 1 & 1 \end{bmatrix}$$

```
(%i39) eigenvectors(%);
(%o39) [[[2],[2]],[[[1,1]]]]
```

Autovalor x=2 (doble)
Autovector = [1,1] = cte [0.5,0.5]

(Nota: sale un solo autovector porque esta matriz NO es diagonalizable)